

**Excel & Business Math**  
**Video/Class Project #31**  
**Night Shift Hours Worked Formula for Payroll Time Sheet**

**Topics**

- 1) Night Shifts Can Have a Time Out that is Smaller than the Time In .....1
- 2) Negative Times are Not Allowed in Excel .....1
- 3) Total Hours Formulas That Works for Day Shift or Night Shift.....2

**1) Night Shifts Can Have a Time Out that is Smaller than the Time In**

As seen in the below picture, the employee Alaniz, Vergie has a Time Out of 3 AM and a Time in of 7 PM. This means that the employee is working the Night Shift where they start before midnight and get off work after midnight. This sort of “Night Shift” Time In and Time Out situation will cause an error with the Total Hours Formula that we learned last video: (Time Out – Time In)\*24.

Employee	Time In	Time Out	Time In as Decimal	Time Out as Decimal
Aasmussen, Kenya	01:00 PM	08:00 PM	0.541666667	0.833333333
Alaniz, Vergie	07:00 PM	03:00 AM	0.791666667	0.125

**2) Negative Times are Not Allowed in Excel**

1. As seen in the below picture, if you create a Time Formula like 3 AM – 7 PM, you get a ##### Error Message because Negative Times are not allowed in Excel.
2. When we create a Time Formula like 3 AM – 7 PM, the underlying decimals will yield a negative decimal of -0.6667, as seen in the picture below.

	A	B	C	D	E	F
7					Negative Times	When Time Out is
8					Not Allowed	Smaller Than Time In,
9						We Get a Negative
				<b>In Our Head we can Make</b>	<b>Later Time - Earlier</b>	<b>Later Time - Earlier</b>
10	<b>Employee</b>	<b>Time In</b>	<b>Time Out</b>	<b>Calculation</b>	<b>Time</b>	<b>Time as decimal</b>
11	Aasmussen, Kenya	01:00 PM	08:00 PM	7 Hours	07:00 AM	0.291666667
12	Alaniz, Vergie	07:00 PM	03:00 AM	8 Hours	#####	-0.666666667
13						
14					Formula in cell E11:	Formula in cell F11:
15					=C11-B11	=C11-B11

### 3) Total Hours Formulas That Works for Day Shift or Night Shift

1. As seen in the below picture, there are three different formulas that we can use to calculate Total Hours as a Proportion of a 24 Hour Day:

	A	B	C	D	E	F	G	H	I	J
7					Negative Times Not Allowed	When Time Out is Smaller Than Time In, We Get a Negative			In Excel Math Formula: TRUE = 1 FALSE = 0	
10	<b>Employee</b>	<b>Time In</b>	<b>Time Out</b>	<b>In Our Head we can Make Calculation</b>	<b>Later Time - Earlier Time</b>	<b>Later Time - Earlier Time as decimal</b>	<b>Night Shift Time with IF</b>	<b>Logical Formula to check if Time Out &lt; Time In</b>	<b>Night Shift Time with Boolean</b>	<b>Night Shift Time with MOD</b>
11	Aasmussen, Kenya	01:00 PM	08:00 PM	7 Hours	07:00 AM	0.291666667	0.291666667	FALSE	0.291666667	0.291666667
12	Alaniz, Vergie	07:00 PM	03:00 AM	8 Hours	#####	-0.666666667	0.333333333	TRUE	0.333333333	0.333333333
14					<b>Formula in cell E11:</b>	<b>Formula in cell F11:</b>	<b>Formula #01 in cell G11:</b>	<b>Formula in cell H11:</b>	<b>Formula #02 in cell I11:</b>	<b>Formula #03 in cell J11:</b>
15					=C11-B11	=C11-B11	=IF(C11<B11,C11-B11+1,C11-B11)	=C11<B11	=C11-B11+(C11<B11)	=MOD(C11-B11,1)

2. As seen in the below picture, there are three different formulas that we can use to calculate Total Hours:

	A	B	C	D	E	F	G	H	I	
1	<b>Payroll Time Sheet with Day &amp; Night Shifts</b>									
3	<b>Employee</b>	<b>Wage per Hour</b>	<b>Time In</b>	<b>Time Out</b>	<b>Hours Worked</b>	<b>Gross Pay</b>		<b>Hours Worked</b>	<b>Hours Worked</b>	
4	Aasmussen, Kenya	\$51.75	01:00 PM	08:00 PM	7	362.25		7	7	
5	Alaniz, Vergie	\$35.30	07:00 PM	03:00 AM	8	282.4		8	8	
6	Biin, Tyrone	\$42.56	10:01 AM	03:47 PM	5.766666667	245.43		5.766666667	5.766666667	
7	Castellano, Odis	\$33.95	03:25 PM	11:43 PM	8.3	281.79		8.3	8.3	
8	Conrad, Fidel	\$42.84	11:24 AM	03:44 PM	4.333333333	185.64		4.333333333	4.333333333	
9	Coughlin, Charmain	\$22.46	02:54 PM	07:46 PM	4.866666667	109.31		4.866666667	4.866666667	
10	Kahn, Cherish	\$51.01	09:32 PM	03:42 AM	6.166666667	314.56		6.166666667	6.166666667	
11	Kellogg, Laureen	\$23.04	06:28 PM	03:47 AM	9.316666667	214.66		9.316666667	9.316666667	
12	Lackey, Willette	\$37.31	09:38 PM	03:53 AM	6.25	233.19		6.25	6.25	
13	Nielsen, Tomi	\$37.66	04:43 PM	02:38 AM	9.916666667	373.46		9.916666667	9.916666667	
14	Quintanilla, Yan	\$48.02	07:19 PM	01:47 AM	6.466666667	310.53		6.466666667	6.466666667	
15	Roldan, Daryl	\$27.81	09:26 PM	01:49 AM	4.383333333	121.9		4.383333333	4.383333333	
16	Salter, Lory	\$48.45	07:05 PM	02:09 AM	7.066666667	342.38		7.066666667	7.066666667	
17	Smithe, Elroy	\$26.08	09:38 PM	02:44 AM	5.1	133.01		5.1	5.1	
18	Toro, Abdi	\$22.12	04:51 PM	08:54 PM	4.05	89.59		4.05	4.05	
20					<b>Formula in cell E4:</b>	<b>Formula in cell F4:</b>		<b>Formula in cell H4:</b>	<b>Formula in cell I4:</b>	
21					=MOD(D4-C4,1)*24	=ROUND(E4*B4,2)		=IF(D4<C4,D4-C4+1,D4-C4)*24	=(D4-C4+(D4<C4))*24	
23					<b>Total</b>	3600.1				